

1.5.4. exercício

1. Escreva um algoritmo que leia notas finais e frequência (S ou I) de alunos de *Introdução à Estatística Computacional*, escreva a nota final, e o conceito correspondente (A, B, C, D, E ou F, de acordo com os valores utilizados na UFMG). Após o último aluno, que terá nota -1, escreva o número de estudantes aprovados e de reprovados na disciplina.

Obs.: Para a determinação do conceito, supor que JÁ exista uma subrotina, CONCEITO (N, F, C), que receba a nota N e a frequência F e retorne o conceito C correspondente.

2. Utilize a subrotina CONCEITO (N, F, C) para criar uma função que receba a nota N e a frequência F e retorne o conceito. Exemplifique a utilização da função criada.

Solução:

1. A solução da primeira parte é apresentada abaixo, partindo da seguinte versão inicial:

```
algoritmo
  defina os tipos das variáveis
  inicialize variáveis
  leia nota, freq
  enquanto nota <>-1 faça
    calcule conceito
    escreva nota, freq, conc
    atualizar número de aprovados e de reprovados
    leia nota, freq
  fim enquanto
  escreva aprov, reprov
fim algoritmo
```

Refinando-se os comandos:

```
ref.: calcule conceito
      CONCEITO(nota, freq, conc)
fim ref.
```

```
ref.: atualizar número de aprovados e de reprovados
      se conc = "F" ou conc = "E" então
        reprov <- reprov + 1
      senão
        aprov <- aprov + 1
      fim se
fim ref.
```

```
ref.: inicialize variáveis  
    reprov <- 0  
    aprov <- 0  
fim ref.
```

```
ref.: defina os tipos das variáveis  
    declare      nota, freq, reprov, aprov numérico  
                conc literal  
fim ref.
```

Inserindo-se os refinamentos nos devidos lugares, tem-se o [algoritmo completo](#):

```
algoritmo  
    declare      nota, freq, reprov, aprov numérico  
                conc literal  
  
    reprov <- 0  
    aprov <- 0  
    leia nota, freq  
    enquanto nota <>-1 faça  
        CONCEITO(nota, freq, conc)  
        escreva nota, conc  
        se conc = "F" ou conc = "E" então  
            reprov <- reprov + 1  
        senão  
            aprov <- aprov + 1  
        fim se  
        leia nota, freq  
    fim enquanto  
    escreva aprov, reprov  
fim algoritmo
```

Observe que a subrotina para cálculo do conceito NÃO precisava ser implementada, pois era suposto que estivesse pronta. Abaixo, sua implementação:

```
subrotina CONCEITO(valor, freq, result)
    {declare variáveis locais}
    declare      valor numérico
                  freq, result caracter
    {calcule conceito}
    se freq = "I" então
        result <- "F"
    senão se valor < 40 então
        result <- "F"
    senão se valor < 59 então
        result <- "E"
    senão se valor < 69 então
        result <- "D"
    senão se valor < 79 então
        result <- "C"
    senão se valor < 89 então
        result <- "B"
    senão
        result <- "A"
    fim se fim se fim se fim se fim se fim se
fim subrotina
```

2. A solução da segunda parte é apresentada abaixo, iniciando-se pela definição da função para cálculo do conceito, a partir das notas e da frequência:

Implementando-se o cálculo do conceito por uma função teríamos:

```
função literal FUNCAOCONCEITO(nota, freq)
    {declare variáveis locais}
    declare      nota numérico
                  freq, result literal
    {calcule conceito}
    CONCEITO(nota, freq, result)
    {retorne resultado}
    FUNCAOCONCEITO <- result
fim subrotina
```

A seguir, um exemplo de utilização da função. Note que a sua chamada é um pouco diferente da chamada da subrotina. Veja como seria o cálculo do conceito:

```
ref.: calcule conceito
      conc <- FUNCAOCONCEITO (nota, freq)
fim ref.
```